ME 201/MTH 281/ME400/CHE400
Contours for Laplace Equation

1.Introduction

In this notebook, we construct contour plots of various solutions of Laplace's equation in a rectangle. The problem considered in section 2 has zero boundary conditions on three edges, and a parabolic distribution on the fourth edge. In section 3, we consider zero boundary conditions on three edges and a constant potential on the fourth edge. In section 4, a movie is constructed which shows in a particular case how the interior solution changes as the boundary conditions are changed.

2. Parabolic Distribution on One Boundary

In this section, we look at the solution of the boundary value problem given below.

$$\frac{\partial^2 \Phi}{\partial x^2} + \frac{\partial^2 \Phi}{\partial y^2} = 0\,,\ 0 < x < a,\ 0 < y < b,$$

$$\text{with } \Phi(0, y) = \Phi(x, 0) = \Phi(a, y) = 0, \text{ and } \Phi(x, b) = \Phi_0 \frac{x}{a}\left(1 - \frac{x}{a}\right).$$

Here $\Phi_0$is a constant. This problem was solved in class by separation of variables. The result is

$$\Phi(x, y) = \sum_{n=1}^{\infty} c_n \frac{\sinh(n\pi y/a)}{\sinh(n\pi b/a)} \sin(n\pi x/a),\quad \text{where } c_n = 0 \text{ for } n \text{ even, } \frac{8\,\Phi_0}{n^3\,\pi^3} \text{ for } n \text{ odd.}$$

We define this series for *Mathematica*. We call the boundary function fbound[x], and its Fourier sine coefficients are called c[n]. The nth term in the series is called term[x,y,n], and the kth partial sum of the series is called phisum[x,y,k].

■ Definition of the Series

```
fbound[x_] := Φ0 * (x / a) * (1 - x / a)

c[n_] := If[OddQ[n], (8 * Φ0) / (n^3 * Pi^3), 0]

term[x_, y_, n_] := N[c[n] * (Sinh[(n * π * y) / a] / Sinh[(n * π * b) / a]) * Sin[(n * π * x) / a]]

phisum[x_, y_, k_] := Sum[term[x, y, n], {n, 1, k}]
```

■ Parameter Values

Now we specify the values of the parameters in the boundary function, and the size of the rectangle. All units are SI.

```
a = 2;  (** m **)

b = 1; (** m **)

Φ0 = 40;  (** volts **)
```

It is useful for plotting purposes to have information on the range of the functions. We specify here the minimum and maximum value of the boundary function. They are called boundmin and boundmax.

```
boundmin := 0;

boundmax := Φ0 / 4;
```

Graphical Check of Sine Series for Boundary Function

We now check that the series correctly represents the boundary function, by plotting both the exact boundary function and the series representation. We use 20 terms in the series, which should be ample given that the convergence is like $1/n^3$ on the boundary. We plot the exact boundary function in red, the series approximation in blue. We first calculate a plot range, which we call plrange.

```
plrange := {boundmin - 0.1 * (boundmax - boundmin), boundmax + 0.1 * (boundmax - boundmin)}

graphbound = Plot[{phisum[x, b, 20], fbound[x]}, {x, 0, a}, PlotRange → plrange, ImageSize → 250,
  AxesLabel → {"x", "boundary function"}, PlotStyle → {RGBColor[0, 0, 1], RGBColor[1, 0, 0]}]
```



The curves coincide, verifying the accuracy of the solution series on the boundary.

Contour Plots

We use *Mathematica*'s contour plot routine to construct a contour plot of the solution. We use 20 terms in the series solution to construct the plot. The list countrange is a calculated set of contour values. We get 11 contours from the minimum boundmin to the maximum boundmax, at equal intervals of potential.

```
countrange := Module[{rng, inc, ans}, rng = boundmax - boundmin; inc = rng / 10;
  ans = {boundmin}; Do[ans = Append[ans, boundmin + j * inc], {j, 1, 10}]; ans]
```

We check this for the present parameter values.

```
countrange

{0, 1, 2, 3, 4, 5, 6, 7, 8, 9, 10}
```

Now we construct the contour plot. We use the contour list constructed above, and we set the aspect ratio to b/a, to keep the picture true to the geometry.

```
topgraph = ContourPlot[phisum[x, y, 20], {x, 0, a}, {y, 0, b}, Contours → countrange,
  AspectRatio → b/a, ImageSize → 250, PlotLabel → "Contours for Solution
of LaPlace Equation"]
```



The 10-volt contour is just a point -- the midpoint of the top boundary -- and doesn't show.

If you use the mouse to pass the cursor over the contour plot, you will see the numerical value of the potential displayed as you cross each contour.

We can eliminate the shading by using the option ContourShading set to False.

```
topgraph2 = ContourPlot[phisum[x, y, 20], {x, 0, a}, {y, 0, b}, Contours → countrange,
  AspectRatio → b/a, ImageSize → 250, ContourShading → False, PlotLabel → "Contours for Solution
of LaPlace Equation"]
```



This gives a clearer view of the lower voltage contours, including the zero contour, which made up of the two sides and the bottom of the box.

3. Constant Potential on One Boundary

We reconsider the above boundary-value problem with the boundary condition now being a constant potential on the upper boundary. The other three sides are still held at zero potential. The new boundary function is

```
fbound[x_] := Φ₀
```

Very few changes are required in what we have defined above. We must redefine the range functions boundmin and boundmax, and we must redefine the Fourier sine coefficients c[n]. We do that now.